
1. [bookmark: _Toc122434485][bookmark: _Toc101278368]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 10.10 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
[bookmark: _Hlk90298041]Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
0. [bookmark: _Toc101278369]Verification Test Summary 
Test Case: 	10.10
ATS Version:	iwd-TTCN3-B2020-09_D22wk12
System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR,
UE used:	Qualcomm SM 8450+SDX65 5G UE
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc101278370]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc97137453][bookmark: _Toc101278371]Change 1
	Function name
	f_TC_10_10_NR5GC_IMS2

	Reason for change
	
1. Resource reservation, PDU Session Modification, will happen during call stablishment. With current implementation it may happen that PDU Session Modification procedure has not finished before Call is released (BYE is sent from SS).
2. After the call is released it is necessary to inform IPCAN and also to handle coordination trigger sent by IPCAN – both done by f_IMS_CC_EndCall.
3. Another coordination trigger to IPCAN is needed to coordinate PDU session release.

	Summary of change
	1. Added a delay of 5s to give some time for PDU Session Modification Complete to NR PTC.
2. Added f_IMS_CC_EndCall in f_TC_10_10_NR5GC_IMS2 after the call is released.
3. Added coordination trigger to IPCAN

	TTCN module
	IMS\NR5GC_10\IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	



Before Change
	  
  function f_TC_10_10_NR5GC_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) MessageBody v_MessageBody;
    var SipUrl v_ContactUrl;
    var boolean v_CondA7 := true;
    
    f_IMS_PTC_Init(PDN_2);

    f_IMS_IMS_WaitForTrigger(OtherPDN);

    f_IMS_TestBody_Set(true);

    // @siclog "Steps 1-8" siclog@
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in steps 1a1 of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // @siclog "Step 9-10" siclog @
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    // EXCEPTION: In parallel to the events described in steps 11-19 below the events specified in 11-13 of Table 4.9.11.2.2-1 of TS 38.508-1 [21] are performed.
    // @siclog "Step 11" siclog@
    // UE sends REGISTER(Step 1 of Annex A.3)
    
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    // @siclog "Step 12" siclog@
    // SS sends 403 Forbidden
    // The SS responds with a failure.
    v_MessageHeader := f_IMS_RegisterResponse_403_MessageHeaderTX(v_RegisterRequest);
    v_MessageHeader.contentType := cs_ContentType(c_ims3gppApplication);
    v_MessageBody := cs_MessageBody_XML_Ims_3gpp(cs_Ims_3gpp_Emergency("anonymous-emergencycall"));
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine403, v_MessageHeader, v_MessageBody)));
    f_IMS_PTC_RegistrationInfo_Reset();
    f_IMS_PTC_Reset();

    f_IMS_PTC_SetRanType(NR); // Ran Type Condition A28
    
    f_IMS_GlobalRelease(oldestProtectedContextOnly);
    
    // @siclog "Step 13" siclog@
    // Check: Does the UE send a correctly composed INVITE request?
    
    // Condition A28 checked
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6);
    
    // Check for INVITE according to Table 10.10.3.3-2
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 13");
    
    //@siclog "Step 14 - 17" siclog@
    //SS sends 100 Trying.(Step 2 of annex A.6)
    //SS sends 180 Ringing
    //SS sends 200 OK.(Step 4 of annex A.6)
    //UE sends ACK.(Step 5 of annex A.6)
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ);
    
    //@siclog "Steps 18-19" siclog@
    // SS sends BYE.(Step 1 of annex A.8)
    //UE sends 200 OK.(Step 2 of annex A.8)
    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrl);
    f_IMS_TestBody_Set(false);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    f_IMS_TCP_CloseHandling(waitAndClose);

  }



After Change
	       function f_TC_10_10_NR5GC_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) MessageBody v_MessageBody;
    var SipUrl v_ContactUrl;
    var boolean v_CondA7 := true;
    
    f_IMS_PTC_Init(PDN_2);

    f_IMS_IMS_WaitForTrigger(OtherPDN);

    f_IMS_TestBody_Set(true);

    // @siclog "Steps 1-8" siclog@
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in steps 1a1 of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // @siclog "Step 9-10" siclog @
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    // EXCEPTION: In parallel to the events described in steps 11-19 below the events specified in 11-13 of Table 4.9.11.2.2-1 of TS 38.508-1 [21] are performed.
    // @siclog "Step 11" siclog@
    // UE sends REGISTER(Step 1 of Annex A.3)
    
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    // @siclog "Step 12" siclog@
    // SS sends 403 Forbidden
    // The SS responds with a failure.
    v_MessageHeader := f_IMS_RegisterResponse_403_MessageHeaderTX(v_RegisterRequest);
    v_MessageHeader.contentType := cs_ContentType(c_ims3gppApplication);
    v_MessageBody := cs_MessageBody_XML_Ims_3gpp(cs_Ims_3gpp_Emergency("anonymous-emergencycall"));
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine403, v_MessageHeader, v_MessageBody)));
    f_IMS_PTC_RegistrationInfo_Reset();
    f_IMS_PTC_Reset();

    f_IMS_PTC_SetRanType(NR); // Ran Type Condition A28
    
    f_IMS_GlobalRelease(oldestProtectedContextOnly);
    
    // @siclog "Step 13" siclog@
    // Check: Does the UE send a correctly composed INVITE request?
    
    // Condition A28 checked
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6);
    
    // Check for INVITE according to Table 10.10.3.3-2
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 13");
    
    //@siclog "Step 14 - 17" siclog@
    //SS sends 100 Trying.(Step 2 of annex A.6)
    //SS sends 180 Ringing
    //SS sends 200 OK.(Step 4 of annex A.6)
    //UE sends ACK.(Step 5 of annex A.6)
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ);
    
    f_Delay(5.0);
    //@siclog "Steps 18-19" siclog@
    // SS sends BYE.(Step 1 of annex A.8)
    //UE sends 200 OK.(Step 2 of annex A.8)
    v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrl);
    f_IMS_CC_EndCall(IPCAN_EmergencyCall_NormalService, IPCAN_SpeechAndEmergencyCall); 

    f_IMS_TestBody_Set(false);
    // Inform NR that call is released (caught by f_NR5GC_EmergencyCallRelease)
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    f_IMS_TCP_CloseHandling(waitAndClose);

  }



1. [bookmark: _Toc101278372]Change 2
	Function name
	f_TC_10_10_NR5GC_IMS1

	Reason for change
	1. The f_IMS_CC_Postamble (if called with TestProcedure parameter) is informing IPCAN to release PDU session, however that was already done from IMS2 and doing it again from IMS1 would lead to runtime error as there are no more PDU sessions.
2. Test purpose dictates that UE registers on V-PLMN and makes the emergency call there.

	Summary of change
	1. f_IMS_CC_Postamble is not called with the TestProcedure parameter.
2. Added new test configuration “IPCAN_EmergencyCall_VPLMN” that uses NR_cell12 which by default is not HPLMN. This change requires also Change 4 and 5 in this document.

Prose CR(R5-222276) is raised during RAN5#95-e.

	TTCN module
	IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	



Before Change
	    function f_TC_10_10_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency call without emergency registration / Failure of registration / Rejected by 403(Forbidden)
    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    //The UE is in test state 1N-A (TS 38.508-1) and registered to IMS.

    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    
    
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release

    
    f_IMS_CC_Postamble(IPCAN_EmergencyCall_NoRegistration);

  }



After Change
	      function f_TC_10_10_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency call without emergency registration / Failure of registration / Rejected by 403(Forbidden)
    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    //The UE is in test state 1N-A (TS 38.508-1) and registered to IMS.
    f_IMS_CC_Preamble(IPCAN_EmergencyCall_VPLMN, IMS_REGISTERED); 
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    
    
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release

    
    f_IMS_CC_Postamble(); 

  }






1. [bookmark: _Toc101278373]Change 3
	Function name
	fl_IP_IMS_Server_SendDL

	Reason for change
	In scenarios where IMS security establishment failed but SIP message needs to be sent in DL nonetheless (emergency call with failed/rejected 401 procedure, e.g. tests 10.9 and 10.10) it is needed to assign UE/NW addresses for routing. 

	Summary of change
	Added routing addresses into fl_IP_IMS_Server_SendDL also when only unprotected context is present.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	



Before Change
	  
    function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,
                                   IMS_RoutingInfo_Type p_RoutingInfo,
                                   IMS_RequestResponse_Type p_RequestResponse,
                                   bitstring p_Data) runs on IP_PTC
  {

…

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {
      v_NW_Address := p_RoutingInfo.NW_Address;
      v_UE_Address := p_RoutingInfo.UE_Address;
    } else if (ispresent(p_ImsServer.RegistrationAddress)) {              /* @sic R5s160735: fl_ImsServer_UeIsRegistered removed sic@ */
      v_NW_Address := p_ImsServer.RegistrationAddress.NW;
      v_UE_Address := p_ImsServer.RegistrationAddress.UE;
    } else {
      FatalError(__FILE__, __LINE__, "missing IP addresses");
    }
…



After Change
	        function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,
                                   IMS_RoutingInfo_Type p_RoutingInfo,
                                   IMS_RequestResponse_Type p_RequestResponse,
                                   bitstring p_Data) runs on IP_PTC
  {

…

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {
      v_NW_Address := p_RoutingInfo.NW_Address;
      v_UE_Address := p_RoutingInfo.UE_Address;
    } else if (ispresent(p_ImsServer.RegistrationAddress)) {              /* @sic R5s160735: fl_ImsServer_UeIsRegistered removed sic@ */
      v_NW_Address := p_ImsServer.RegistrationAddress.NW;
      v_UE_Address := p_ImsServer.RegistrationAddress.UE;
    } else if (ispresent(p_ImsServer.UnprotectedContext.TcpServer)) {              
      v_NW_Address := p_ImsServer.UnprotectedContext.TcpServer.Local.IpAddr;
      v_UE_Address := p_ImsServer.UnprotectedContext.TcpServer.Remote.IpAddr;
    } else {
      FatalError(__FILE__, __LINE__, "missing IP addresses");
    }
…




1. [bookmark: _Toc101278374]Change 4
	Function name
	f_NR5GC_IPCAN_MainLoop

	Reason for change
	To accommodate Change 2 bullet 2

	Summary of change
	Added new test configuration “IPCAN_EmergencyCall_VPLMN” that uses NR_cell12 which by default is not HPLMN.

	TTCN module
	PART3_IPCAN\IPCAN_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	    function f_NR5GC_IPCAN_MainLoop() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_CellId := tsc_IMS_NR5GC_CellId;
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var IMS_IPCAN_Command_Type v_Command;
    var IMS_TestConfiguration_Type v_TestConfiguration;
    var TriggerResult_Type v_StartProcedureResult;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var integer v_PdnIndex;
    var boolean v_ConnectionEstablished := false;
    var boolean v_IPCAN_Registered := false;   // @sic R5s211495 sic@
    var boolean v_Done := false;

    while (not v_Done) {
      alt {
        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS1;
        }
        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS2;
        }
        // @sic R5s211495 sic@
        [v_ConnectionEstablished]SRB.receive(car_NR_SRB_NasPdu_IND(v_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered)))) -> value v_ReceivedAsp {
          v_IPCAN_Registered := f_NR5GC_IPCAN_AfterIMSDeregister(v_CellId, v_ReceivedAsp);
          repeat;
        }
      }
      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;
      select (v_Command.Name) {
        case (IPCAN_INIT) {
          f_NR5GC_IPCAN_Init(IMS[v_PdnIndex], v_CellId);
          v_TestConfiguration := v_Command.TestConfiguration;



        }
        case (IPCAN_PROVIDELOCATIONINFORMATION) {
          f_NR5GC_IPCAN_LocationInformation(IMS[v_PdnIndex], v_CellId);
        }
        case (IPCAN_STARTPROCEDURE) {
          v_StartProcedureResult := f_NR5GC_IPCAN_StartProcedure(IMS[v_PdnIndex],
                                                                 v_CellId,
                                                                 v_Command.TestProcedure,
                                                                 v_ConnectionEstablished);
          v_ConnectionEstablished := true;
          if ( v_Command.TestProcedure == IPCAN_InitialRegistration) {    // @sic R5s211495 sic@
             v_IPCAN_Registered := true;
          }
        }
        case (IPCAN_ENDPROCEDURE) {
          if (v_TestConfiguration == IPCAN_SpeechAndEmergencyCall and isvalue(v_Command.TestConfiguration)) { // @sic R5-216220 sic@
            // Just end the procedure as there's 2 calls ongoing, don't release the connection
            f_NR5GC_IPCAN_EndCall(IMS[v_PdnIndex], v_CellId, valueof(v_Command.TestConfiguration), v_Command.TestProcedure, v_StartProcedureResult);
          } else {
            f_NR5GC_IPCAN_ConnectionRelease(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure, v_StartProcedureResult);
            /* => RRC connection is released */
            v_ConnectionEstablished := false;
          }
        }
        case (IPCAN_RELEASE) {
          f_NR5GC_IPCAN_Release(IMS[v_PdnIndex], v_CellId, v_IPCAN_Registered); // @sic R5s211495 sic@
          v_Done := true;
        }
        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */
          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_NR_IMS_IpcanInfo()));
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid command");
        }
      }
    }
  }




After Change
	  const NR_CellId_Type tsc_IMS_NR5GC_CellId_VPLMN := nr_Cell12; 

…

  function f_NR5GC_IPCAN_MainLoop() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_CellId := tsc_IMS_NR5GC_CellId;
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var IMS_IPCAN_Command_Type v_Command;
    var IMS_TestConfiguration_Type v_TestConfiguration;
    var TriggerResult_Type v_StartProcedureResult;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var integer v_PdnIndex;
    var boolean v_ConnectionEstablished := false;
    var boolean v_IPCAN_Registered := false;   // @sic R5s211495 sic@
    var boolean v_Done := false;

    while (not v_Done) {
      alt {
        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS1;
        }
        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {
          v_PdnIndex := tsc_Index_IMS2;
        }
        // @sic R5s211495 sic@
        [v_ConnectionEstablished]SRB.receive(car_NR_SRB_NasPdu_IND(v_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered)))) -> value v_ReceivedAsp {
          v_IPCAN_Registered := f_NR5GC_IPCAN_AfterIMSDeregister(v_CellId, v_ReceivedAsp);
          repeat;
        }
      }
      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;
      select (v_Command.Name) {
        case (IPCAN_INIT) {

          v_TestConfiguration := v_Command.TestConfiguration;
          if (v_Command.TestConfiguration == IPCAN_EmergencyCall_VPLMN) {v_CellId := tsc_IMS_NR5GC_CellId_VPLMN}
          f_NR5GC_IPCAN_Init(IMS[v_PdnIndex], v_CellId);

        }
        case (IPCAN_PROVIDELOCATIONINFORMATION) {
          f_NR5GC_IPCAN_LocationInformation(IMS[v_PdnIndex], v_CellId);
        }
        case (IPCAN_STARTPROCEDURE) {
          v_StartProcedureResult := f_NR5GC_IPCAN_StartProcedure(IMS[v_PdnIndex],
                                                                 v_CellId,
                                                                 v_Command.TestProcedure,
                                                                 v_ConnectionEstablished);
          v_ConnectionEstablished := true;
          if ( v_Command.TestProcedure == IPCAN_InitialRegistration) {    // @sic R5s211495 sic@
             v_IPCAN_Registered := true;
          }
        }
        case (IPCAN_ENDPROCEDURE) {
          if (v_TestConfiguration == IPCAN_SpeechAndEmergencyCall and isvalue(v_Command.TestConfiguration)) { // @sic R5-216220 sic@
            // Just end the procedure as there's 2 calls ongoing, don't release the connection
            f_NR5GC_IPCAN_EndCall(IMS[v_PdnIndex], v_CellId, valueof(v_Command.TestConfiguration), v_Command.TestProcedure, v_StartProcedureResult);
          } else {
            f_NR5GC_IPCAN_ConnectionRelease(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure, v_StartProcedureResult);
            /* => RRC connection is released */
            v_ConnectionEstablished := false;
          }
        }
        case (IPCAN_RELEASE) {
          f_NR5GC_IPCAN_Release(IMS[v_PdnIndex], v_CellId, v_IPCAN_Registered); // @sic R5s211495 sic@
          v_Done := true;
        }
        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */
          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_NR_IMS_IpcanInfo()));
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid command");
        }
      }
    }
  }




1. [bookmark: _Toc101278375]Change 5
	Type name
	IMS_TestConfiguration_Type

	Reason for change
	To accommodate Change 2 bullet 2

	Summary of change
	Added new test configuration “IPCAN_EmergencyCall_VPLMN” that uses NR_cell12 which by default is not HPLMN.

	TTCN module
	IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	



Before Change
	    type enumerated IMS_TestConfiguration_Type {  /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    IPCAN_SignallingOnly,                       /* EUTRA: default DRB is used only */
    IPCAN_SpeechCall,                           /* EUTRA: one dedicated UM bearer; for normal speech calls and emergency call for limited services */
    IPCAN_VideoCall,                            /* EUTRA: two dedicated UM bearers */
    IPCAN_EmergencyCall,                        /* EUTRA: second default bearer (AM) and one dedicated UM bearer */
    IPCAN_SpeechAndEmergencyCall,               /* EUTRA: default bearer + dedicated bearer for normal speech call and another default bearer + dedicated bearer for emergency call */
    IPCAN_XCAP,                                 /* EUTRA: second default bearer (AM) for second PDN used for XCAP signalling */
    IPCAN_DataCall                              /* MCData: one dedicated UM bearer */

  };




After Change
	  type enumerated IMS_TestConfiguration_Type {  /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    IPCAN_SignallingOnly,                       /* EUTRA: default DRB is used only */
    IPCAN_SpeechCall,                           /* EUTRA: one dedicated UM bearer; for normal speech calls and emergency call for limited services */
    IPCAN_VideoCall,                            /* EUTRA: two dedicated UM bearers */
    IPCAN_EmergencyCall,                        /* EUTRA: second default bearer (AM) and one dedicated UM bearer */
    IPCAN_SpeechAndEmergencyCall,               /* EUTRA: default bearer + dedicated bearer for normal speech call and another default bearer + dedicated bearer for emergency call */
    IPCAN_XCAP,                                 /* EUTRA: second default bearer (AM) for second PDN used for XCAP signalling */
    IPCAN_DataCall,                              /* MCData: one dedicated UM bearer */
    IPCAN_EmergencyCall_VPLMN
  };





1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc101278376]Branches executed 
On Anritsu, the test case was executed on NR band n41 using NR ciphering nea1 and integrity algorithm nia1.
3 [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc101278377]Execution Log Files 
3.1 [bookmark: _Toc97137458][bookmark: _Toc101278378]Qualcomm SM 8450+SDX65 5G UE
[bookmark: _Hlk90552893]The Qualcomm SDX UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_10_10_LOG.xml
· PICS/PIXIT parameter file:
TC_10_10_PIXIT.txt


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc101278379]References 
	[1]
	R5s220612 - Supporting information for addition of IMS 5GS emergency call  test case 10.10





